
Overview ASP II

• Modeling
1 Basic Modeling
2 Methodology

3 Motivation
4 Core language

Modeling: Overview

1 Basic Modeling

2 Methodology

Modeling and Interpreting

Problem

Logic Program

Solution

Stable Models

Modeling

• For solving a problem class C for a problem instance I,
encode

1 the problem instance I as a set PI of facts and
2 the problem class C as a set PC of rules

such that the solutions to C for I can be (polynomially) extracted
from the stable models of PI ∪ PC

• PI is (still) called problem instance
• PC is often called the problem encoding

• An encoding PC is uniform, if it can be used to solve all its
problem instances
That is, PC encodes the solutions to C for any set PI of facts

1 Basic Modeling

2 Methodology

Basic methodology

Methodology
Generate and Test (or: Guess and Check)

Generator Generate potential stable model candidates
(typically through non-deterministic constructs)

Tester Eliminate invalid candidates
(typically through integrity constraints)

Nutshell
Logic program = Data + Generator + Tester ( + Optimizer)

Outline

1 Basic Modeling

2 Methodology
Satisfiability
Queens
Traveling Salesperson

Satisfiability testing

• Problem Instance: A propositional formula φ in CNF
• Problem Class: Is there an assignment of propositional variables to true

and false such that a given formula φ is true

• Example: Consider formula

(a ∨ ¬b) ∧ (¬a ∨ b)

• Logic Program:

Generator Tester Stable models
{ a, b } ← ← not a, b

← a, not b
X1 = {a, b}
X2 = {}

The n-Queens Problem

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 0Z0Z0
1 Z0Z0Z

1 2 3 4 5

• Place n queens on an n× n
chess board

• Queens must not attack one
another

Q Q Q

Q Q

Defining the Field

queens.lp

row(1..n).
col(1..n).

• Create file queens.lp

• Define the field
– n rows
– n columns

Defining the Field

Running . . .

$ gringo queens.lp --const n=5 | clasp
Answer: 1
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5)
SATISFIABLE

Models : 1
Time : 0.000
Prepare : 0.000
Prepro. : 0.000
Solving : 0.000

Placing some Queens

queens.lp

row(1..n).
col(1..n).
{ queen(I,J) : row(I), col(J) }.

• Guess a solution candidate

by placing some queens on the board

TU Dresden PSSAI slide 21 of 101



Placing some Queens

Running . . .

$ gringo queens.lp --const n=5 | clasp 3
Answer: 1
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5)
Answer: 2
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5) queen(1,1)
Answer: 3
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5) queen(2,1)
SATISFIABLE

Models : 3+
...

Placing some Queens: Answer 1

Answer 1

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 0Z0Z0
1 Z0Z0Z

1 2 3 4 5

Placing some Queens: Answer 2

Answer 2

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 0Z0Z0
1 L0Z0Z

1 2 3 4 5

Placing some Queens: Answer 3

Answer 3

5 Z0Z0Z
4 0Z0Z0
3 Z0Z0Z
2 QZ0Z0
1 Z0Z0Z

1 2 3 4 5

Placing n Queens

queens.lp

row(1..n).
col(1..n).
{ queen(I,J) : row(I), col(J) }.
:- not n { queen(I,J) } n.

• Place exactly n queens on the board

Placing n Queens

Running . . .

$ gringo queens.lp --const n=5 | clasp 2
Answer: 1
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5) \
queen(5,1) queen(4,1) queen(3,1) \
queen(2,1) queen(1,1)
Answer: 2
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5) \
queen(1,2) queen(4,1) queen(3,1) \
queen(2,1) queen(1,1)
...

Placing n Queens: Answer 1

Answer 1

5 L0Z0Z
4 QZ0Z0
3 L0Z0Z
2 QZ0Z0
1 L0Z0Z

1 2 3 4 5

Placing n Queens: Answer 2

Answer 2

5 Z0Z0Z
4 QZ0Z0
3 L0Z0Z
2 QZ0Z0
1 LQZ0Z

1 2 3 4 5

Horizontal and Vertical Attack

queens.lp

row(1..n).
col(1..n).
{ queen(I,J) : row(I), col(J) }.
:- not n { queen(I,J) } n.
:- queen(I,J), queen(I,J’), J != J’.

:- queen(I,J), queen(I’,J), I != I’.

• Forbid horizontal attacks

• Forbid vertical attacks

Horizontal and Vertical Attack

Running . . .

$ gringo queens.lp --const n=5 | clasp
Answer: 1
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5) \
queen(5,5) queen(4,4) queen(3,3) \
queen(2,2) queen(1,1)
...

Horizontal and Vertical Attack: Answer 1

Answer 1

5 Z0Z0L
4 0Z0L0
3 Z0L0Z
2 0L0Z0
1 L0Z0Z

1 2 3 4 5

Diagonal Attack

queens.lp

row(1..n).
col(1..n).
{ queen(I,J) : row(I), col(J) }.
:- not n { queen(I,J) } n.
:- queen(I,J), queen(I,J’), J != J’.
:- queen(I,J), queen(I’,J), I != I’.
:- queen(I,J), queen(I’,J’), (I,J) != (I’,J’), I-J ==

I’-J’.
:- queen(I,J), queen(I’,J’), (I,J) != (I’,J’), I+J ==

I’+J’.

• Forbid diagonal attacks

Diagonal Attack

Running . . .

$ gringo queens.lp --const n=5 | clasp
Answer: 1
row(1) row(2) row(3) row(4) row(5) \
col(1) col(2) col(3) col(4) col(5) \
queen(4,5) queen(1,4) queen(3,3) queen(5,2) queen(2,1)
SATISFIABLE

Models : 1+
Time : 0.000
Prepare : 0.000
Prepro. : 0.000
Solving : 0.000

Diagonal Attack: Answer 1

Answer 1

5 ZQZ0Z
4 0Z0ZQ
3 Z0L0Z
2 QZ0Z0
1 Z0ZQZ

1 2 3 4 5

Optimizing

queens-opt.lp

1 { queen(I,1..n) } 1 :- I = 1..n.
1 { queen(1..n,J) } 1 :- J = 1..n.
:- 2 { queen(D-J,J) }, D = 2..2*n.
:- 2 { queen(D+J,J) }, D = 1-n..n-1.

• Encoding can be optimized
• Much faster to solve

TU Dresden PSSAI slide 37 of 101



And sometimes it rocks
$ clingo -c n=5000 queens-opt-diag.lp -config=jumpy -q -stats=3
clingo version 4.1.0
Solving...
SATISFIABLE

Models : 1+
Time : 3758.143s (Solving: 1905.22s 1st Model: 1896.20s Unsat: 0.00s)
CPU Time : 3758.320s

Choices : 288594554
Conflicts : 3442 (Analyzed: 3442)
Restarts : 17 (Average: 202.47 Last: 3442)
Model-Level : 7594728.0
Problems : 1 (Average Length: 0.00 Splits: 0)
Lemmas : 3442 (Deleted: 0)

Binary : 0 (Ratio: 0.00%)
Ternary : 0 (Ratio: 0.00%)
Conflict : 3442 (Average Length: 229056.5 Ratio: 100.00%)
Loop : 0 (Average Length: 0.0 Ratio: 0.00%)
Other : 0 (Average Length: 0.0 Ratio: 0.00%)

Atoms : 75084857 (Original: 75069989 Auxiliary: 14868)
Rules : 100129956 (1: 50059992/100090100 2: 39990/29856 3: 10000/10000)
Bodies : 25090103
Equivalences : 125029999 (Atom=Atom: 50009999 Body=Body: 0 Other: 75020000)
Tight : Yes
Variables : 25024868 (Eliminated: 11781 Frozen: 25000000)
Constraints : 66664 (Binary: 35.6% Ternary: 0.0% Other: 64.4%)

Backjumps : 3442 (Average: 681.19 Max: 169512 Sum: 2344658)
Executed : 3442 (Average: 681.19 Max: 169512 Sum: 2344658 Ratio: 100.00%)
Bounded : 0 (Average: 0.00 Max: 0 Sum: 0 Ratio: 0.00%)

Traveling Salesperson

node(1..6).

edge(1,(2;3;4)). edge(2,(4;5;6)). edge(3,(1;4;5)).
edge(4,(1;2)). edge(5,(3;4;6)). edge(6,(2;3;5)).

cost(1,2,2). cost(1,3,3). cost(1,4,1).
cost(2,4,2). cost(2,5,2). cost(2,6,4).
cost(3,1,3). cost(3,4,2). cost(3,5,2).
cost(4,1,1). cost(4,2,2).
cost(5,3,2). cost(5,4,2). cost(5,6,1).
cost(6,2,4). cost(6,3,3). cost(6,5,1).

Traveling Salesperson

edge(X,Y) :- cost(X,Y,_).

Traveling Salesperson

1 { cycle(X,Y) : edge(X,Y) } 1 :- node(X).
1 { cycle(X,Y) : edge(X,Y) } 1 :- node(Y).

reached(Y) :- cycle(1,Y).
reached(Y) :- cycle(X,Y), reached(X).

:- node(Y), not reached(Y).

#minimize { C,X,Y : cycle(X,Y), cost(X,Y,C) }.

Language: Overview

3 Motivation

4 Core language

Outline

3 Motivation

4 Core language

Basic language extensions

• The expressiveness of a language can be enhanced by introducing
new constructs

• To this end, we must address the following issues:
– What is the syntax of the new language construct?
– What is the semantics of the new language construct?
– How to implement the new language construct?

• A way of providing semantics is to furnish a translation removing the new
constructs, eg. classical negation

• This translation might also be used for implementing the language
extension

Outline

Outline

3 Motivation

4 Core language
Integrity constraint
Choice rule
Cardinality rule
Weight rule

Integrity constraint

• Idea Eliminate unwanted solution candidates
• Syntax An integrity constraint is of the form

← a1, . . . , am, not am+1, . . . , not an

where 0 ≤ m ≤ n and each ai is an atom for 1 ≤ i ≤ n

• Example :- edge(3,7), color(3,red), color(7,red).

• Embedding The above integrity constraint can be turned into the normal
rule

x← a1, . . . , am, not am+1, . . . , not an, not x

where x is a new symbol, that is, x 6∈ A.

Outline

3 Motivation

4 Core language

Choice rule

• Idea Choices over subsets
• Syntax A choice rule is of the form

{a1, . . . , am} ← am+1, . . . , an, not an+1, . . . , not ao

where 0 ≤ m ≤ n ≤ o and each ai is an atom for 1 ≤ i ≤ o

• Informal meaning If the body is satisfied by the stable model at hand,
then any subset of {a1, . . . , am} can be included in the stable model

• Example
{ buy(pizza); buy(wine); buy(corn) } :- at(grocery).

• Another Example P = {{a} ← b, b←} has two stable models: {b}
and {a, b}

Embedding in normal rules

• A choice rule of form

{a1, . . . , am} ← am+1, . . . , an, not an+1, . . . , not ao

can be translated into 2m + 1 normal rules

b ← am+1, . . . , an, not an+1, . . . , not ao

a1 ← b, not a′1 . . . am ← b, not a′m
a′1 ← not a1 . . . a′m ← not am

by introducing new atoms b, a′1, . . . , a′m.

Outline

3 Motivation

4 Core language
Integrity constraint
Choice rule
Cardinality rule
Weight rule

Cardinality rule

• Idea Control (lower) cardinality of subsets
• Syntax A cardinality rule is the form

a0 ← l { a1, . . . , am, not am+1, . . . , not an }

where 0 ≤ m ≤ n and each ai is an atom for 1 ≤ i ≤ n;
l is a non-negative integer.

• Informal meaning The head atom belongs to the stable model,
if at least l elements of the body are included in the stable model

• Note l acts as a lower bound on the body

• Example
pass(c42) :- 2 { pass(a1); pass(a2); pass(a3) }.

• Another Example P = {a← 1{b, c}, b←} has stable model {a, b}

• Idea Control (lower) cardinality of subsets
• Syntax A cardinality rule is the form

a0 ← l { a1, . . . , am, not am+1, . . . , not an }

where 0 ≤ m ≤ n and each ai is an atom for 1 ≤ i ≤ n;
l is a non-negative integer.

• Informal meaning The head atom belongs to the stable model,
if at least l elements of the body are included in the stable model

• Note l acts as a lower bound on the body

• Example

Embedding in normal rules

• Replace each cardinality rule

a0 ← l { a1, . . . , am, not am+1, . . . , not an }

by a0 ← ctr(1, l)

where atom ctr(i, j) represents the fact that at least j of the literals having
an equal or greater index than i, are in a stable model

• The definition of ctr/2 is given for 0 ≤ k ≤ l by the rules

ctr(i, k+1) ← ctr(i + 1, k), ai
ctr(i, k) ← ctr(i + 1, k) for 1 ≤ i ≤ m

ctr(j, k+1) ← ctr(j + 1, k), not aj
ctr(j, k) ← ctr(j + 1, k) for m + 1 ≤ j ≤ n

ctr(n + 1, 0) ←

An example

• Program {a←, c← 1 {a, b}} has the stable model {a, c}

• Translating the cardinality rule yields the rules

a ← c ← ctr(1, 1)
ctr(1, 2) ← ctr(2, 1), a
ctr(1, 1) ← ctr(2, 1)
ctr(2, 2) ← ctr(3, 1), b
ctr(2, 1) ← ctr(3, 1)
ctr(1, 1) ← ctr(2, 0), a
ctr(1, 0) ← ctr(2, 0)
ctr(2, 1) ← ctr(3, 0), b
ctr(2, 0) ← ctr(3, 0)
ctr(3, 0) ←

having stable model {a, ctr(3, 0), ctr(2, 0), ctr(1, 0), ctr(1, 1), c}

. . . and vice versa

• A normal rule
a0 ← a1, . . . , am, not am+1, . . . , not an

can be represented by the cardinality rule

a0 ← n {a1, . . . , am, not am+1, . . . , not an}

TU Dresden PSSAI slide 84 of 101



Cardinality rules with upper bounds

• A rule of the form

a0 ← l { a1, . . . , am, not am+1, . . . , not an } u (1)

where 0 ≤ m ≤ n and each ai is an atom for 1 ≤ i ≤ n;
l and u are non-negative integers

stands for

a0 ← b, not c
b ← l { a1, . . . , am, not am+1, . . . , not an }
c ← u+1 { a1, . . . , am, not am+1, . . . , not an }

where b and c are new symbols

• Note The single constraint in the body of the cardinality rule (1) is referred
to as a cardinality constraint

Cardinality constraints

• Syntax A cardinality constraint is of the form

l { a1, . . . , am, not am+1, . . . , not an } u

where 0 ≤ m ≤ n and each ai is an atom for 1 ≤ i ≤ n;
l and u are non-negative integers

• Informal meaning A cardinality constraint is satisfied by a stable model X,
if the number of its contained literals satisfied by X is between l and u
(inclusive)

• In other words, if

l ≤ | ({a1, . . . , am} ∩ X) ∪ ({am+1, . . . , an} \ X) | ≤ u

Cardinality constraints as heads

• A rule of the form

l {a1, . . . , am, not am+1, . . . , not an} u← an+1, . . . , ao, not ao+1, . . . , not ap

where 0 ≤ m ≤ n ≤ o ≤ p and each ai is an atom for 1 ≤ i ≤ p;
l and u are non-negative integers

stands for

b ← an+1, . . . , ao, not ao+1, . . . , not ap
{a1, . . . , am} ← b

c ← l {a1, . . . , am, not am+1, . . . , not an} u
← b, not c

where b and c are new symbols

• Example 1{ color(v42,red); color(v42,green); color(v42,blue) }1.

Outline

3 Motivation

4 Core language
Integrity constraint
Choice rule
Cardinality rule
Weight rule

Weight rule

• Syntax A weight rule is the form

a0 ← l { w1 : a1, . . . , wm : am, wm+1 : not am+1, . . . , wn : not an }

where 0 ≤ m ≤ n and each ai is an atom;
l and wi are integers for 1 ≤ i ≤ n

• A weighted literal wi : `i associates each literal `i with a weight wi

• Note A cardinality rule is a weight rule where wi = 1 for 0 ≤ i ≤ n

Weight constraints

• Syntax A weight constraint is of the form

l { w1 : a1, . . . , wm : am, wm+1 : not am+1, . . . , wn : not an } u

where 0 ≤ m ≤ n and each ai is an atom;
l, u and wi are integers for 1 ≤ i ≤ n

• Meaning A weight constraint is satisfied by a stable model X, if

l ≤
(∑

1≤i≤m,ai∈X wi +
∑

m<i≤n,ai 6∈X wi

)
≤ u

• Note (Cardinality and) weight constraints amount to constraints on (count
and) sum aggregate functions

• Example
10 { 4:course(db); 6:course(ai); 8:course(project); 3:course(xml) } 20

• See also: http://potassco.sourceforge.net

